

Projeto ESTOQUE

Documentação técnica completa
WINDEV 2025 · Update 1 · 03/09/2026

Sumário: 1. Código · 2. Interfaces · 3. Queries

Projeto ESTOQUE — Documentação técnica: Código

WINDEV 2025 · Update 1 · Gerado em 03/09/2026 · Idioma do projeto: pt-BR · Escopo: procedures globais, classes e eventos

Código do projeto

Inicialização do projeto

```
// Conexão com o HFSQL Client/Server; parâmetros vêm do arquivo estoque.ini
gsConexao is Connection
gsConexao.Provider = hAccessHFClientServer
gsConexao.Server = INIRead("HFSQL", "Servidor", "localhost:4900", fExeDir() +
"\estoque.ini")
gsConexao.Database = "ESTOQUE"
gsConexao.User = INIRead("HFSQL", "Usuario", "admin", fExeDir() + "\estoque.ini")
HChangeConnection("", gsConexao)
IF NOT HOpenConnection(gsConexao) THEN
    Error("Não foi possível conectar ao servidor HFSQL.", HErrorInfo())
EndProgram()
END
gnUsuarioLogado is int = 0
gsPerfil is string = ""
```

Conjunto de procedures globais: Regras

CalculaDesconto

```
// Regra comercial: desconto máximo por tipo de cliente.
// Cliente comum (C): até 15%. Cliente especial (E): até 25%.
// Vendedor não pode ultrapassar o teto; quem ultrapassa recebe erro e a venda não
fecha.
PROCEDURE CalculaDesconto(nSubtotal is currency, nPercentual is numeric, sTipoCliente is
string): currency
nTeto is numeric
SWITCH sTipoCliente
    CASE "E": nTeto = 25
    OTHER CASE: nTeto = 15
END
IF nPercentual > nTeto THEN
    Error("Desconto de " + nPercentual + "% ultrapassa o máximo de " + nTeto + "% para
este cliente.")
    RETURN -1
END
RETURN Round(nSubtotal * nPercentual / 100, 2)
```

ValidaEstoque

```
// Regra de estoque: não vende o que não tem no depósito escolhido.
// Retorna a quantidade disponível ou -1 quando o produto não tem saldo no depósito.
PROCEDURE ValidaEstoque(nIDProduto is int, nIDDeposito is int, nQuantidade is numeric):
boolean
HReadSeekFirst(SALDO, IDPRODUTO_IDDEPOSITO, [nIDProduto, nIDDeposito])
IF NOT HFound(SALDO) THEN
    Error("Produto sem saldo cadastrado neste depósito.")
    RETURN False
END
IF SALDO.QUANTIDADE < nQuantidade THEN
    Error("Estoque insuficiente. Disponível: " + NumToString(SALDO.QUANTIDADE, "12.3f"))
    RETURN False
```

```
END
RETURN True
```

BaixaEstoque

```
// Baixa transacional: ou baixa todos os itens da venda, ou nenhum.
PROCEDURE BaixaEstoque(nIDVenda is int): boolean
HTransactionStart(gsConexao)
HReadSeekFirst(ITEMVENDA, IDVENDA, nIDVenda)
WHILE HFound(ITEMVENDA)
    HReadSeekFirst(SALDO, IDPRODUTO_IDDEPOSITO, [ITEMVENDA.IDPRODUTO, VENDA.IDDEPOSITO])
    IF NOT HFound(SALDO) OR SALDO.QUANTIDADE < ITEMVENDA.QUANTIDADE THEN
        HTransactionCancel(gsConexao)
        Error("Baixa cancelada: item " + ITEMVENDA.IDPRODUTO + " sem saldo.")
        RETURN False
    END
    SALDO.QUANTIDADE -= ITEMVENDA.QUANTIDADE
    HModify(SALDO)
    HReadNext(ITEMVENDA, IDVENDA)
END
VENDA.STATUS = "F"
HModify(VENDA)
HTransactionEnd(gsConexao)
RETURN True
```

CalculaJurosAtraso

```
// Regra financeira: 2% ao mês pro rata die sobre o valor do título, a partir do dia
seguinte ao vencimento.
// Título pago no vencimento não tem juros.
PROCEDURE CalculaJurosAtraso(nValor is currency, dVencimento is Date, dPagamento is
Date): currency
nDias is int = DateDifference(dVencimento, dPagamento)
IF nDias <= 0 THEN RETURN 0
RETURN Round(nValor * 0.02 / 30 * nDias, 2)
```

ValidaCPF

```
// Validação de CPF pelos dois dígitos verificadores; rejeita sequências repetidas.
PROCEDURE ValidaCPF(sCPF is string): boolean
sCPF = NoSpace(Replace(Replace(sCPF, ".", ""), "-", ""))
IF Length(sCPF) <> 11 THEN RETURN False
IF sCPF = RepeatString(sCPF[[1]], 11) THEN RETURN False
nSoma, i are int
FOR i = 1 TO 9
    nSoma += Val(sCPF[[i]]) * (11 - i)
END
nDV1 is int = modulo(nSoma * 10, 11)
IF nDV1 = 10 THEN nDV1 = 0
IF nDV1 <> Val(sCPF[[10]]) THEN RETURN False
nSoma = 0
FOR i = 1 TO 10
    nSoma += Val(sCPF[[i]]) * (12 - i)
END
nDV2 is int = modulo(nSoma * 10, 11)
IF nDV2 = 10 THEN nDV2 = 0
RETURN nDV2 = Val(sCPF[[11]])
```

Janela WIN_Venda — eventos

Inicialização de WIN_Venda

```

VENDA.DATA = Today()
VENDA.VENDEDOR = gsUsuarioNome
VENDA.STATUS = "A"
TABLE_Itens.DeleteAll()
EDT_Subtotal = 0
EDT_Desconto = 0
EDT_Total = 0

```

Clique em BTN_AgregarItem

```

IF COMBO_Produto = 0 THEN
    Error("Escolha um produto.")
    ReturnToCapture(COMBO_Produto)
END
IF EDT_Quantidade <= 0 THEN
    Error("Quantidade deve ser maior que zero.")
    ReturnToCapture(EDT_Quantidade)
END
IF NOT ValidaEstoque(COMBO_Produto, COMBO_Deposito, EDT_Quantidade) THEN
    ReturnToCapture(EDT_Quantidade)
END
HReadSeekFirst(PRODUTO, IDPRODUTO, COMBO_Produto)
TableAddLine(TABLE_Itens, PRODUTO.CODIGO, PRODUTO.DESCRICAO, EDT_Quantidade,
PRODUTO.PRECO_VENDA, Round(EDT_Quantidade * PRODUTO.PRECO_VENDA, 2))
RecalculaTotais()

```

Procedure local RecalculaTotais

```

PROCEDURE RecalculaTotais()
nSub is currency = 0
FOR EACH ROW OF TABLE_Itens
    nSub += TABLE_Itens.COL_TotalItem
END
EDT_Subtotal = nSub
nDesc is currency = CalculaDesconto(nSub, EDT_PercDesconto, CLIENTE.TIPO)
IF nDesc < 0 THEN
    EDT_PercDesconto = 0
    nDesc = 0
END
EDT_Desconto = nDesc
EDT_Total = nSub - nDesc

```

Clique em BTN_Fechar

```

IF TABLE_Itens.Count = 0 THEN
    Error("A venda não tem itens.")
    RETURN
END
IF EDT_Total > CLIENTE.LIMITE_CREDITO AND CLIENTE.TIPO = "C" THEN
    IF NOT YesNo(No, "Total acima do limite de crédito do cliente. Fechar mesmo assim?")
    THEN RETURN
END
ScreenToFile(WIN_Venda)
HAdd(VENDA)
FOR EACH ROW OF TABLE_Itens
    ITEMVENDA.IDVENDA = VENDA.IDVENDA
    HReadSeekFirst(PRODUTO, CODIGO, TABLE_Itens.COL_Codigo)
    ITEMVENDA.IDPRODUTO = PRODUTO.IDPRODUTO
    ITEMVENDA.QUANTIDADE = TABLE_Itens.COL_Quantidade
    ITEMVENDA.PRECO_UNITARIO = TABLE_Itens.COL_Preco
    ITEMVENDA.TOTAL_ITEM = TABLE_Itens.COL_TotalItem
    HAdd(ITEMVENDA)
END

```

```

IF BaixaEstoque(VENDA.IDVENDA) THEN
    GeraTitulos(VENDA.IDVENDA, EDT_Total, COMBO_Parcelas)
    Info("Venda " + VENDA.IDVENDA + " fechada.")
    Close()
END

```

Procedure global GeraTitulos

```

// Parcelas iguais, vencendo a cada 30 dias; a diferença de arredondamento vai para a
última.
PROCEDURE GeraTitulos(nIDVenda is int, nTotal is currency, nParcelas is int)
nValorParcela is currency = Round(nTotal / nParcelas, 2)
nAcumulado is currency = 0
FOR i = 1 TO nParcelas
    TITULO.IDVENDA = nIDVenda
    TITULO.VENCIMENTO = DateSys() + 30 * i
    IF i = nParcelas THEN
        TITULO.VALOR = nTotal - nAcumulado
    ELSE
        TITULO.VALOR = nValorParcela
        nAcumulado += nValorParcela
    END
    HAdd(TITULO)
END

```

Janela WIN_Cliente — eventos

Saída de EDT_CPF

```

IF NOT ValidaCPF(EDT_CPF) THEN
    Error("CPF inválido.")
    ReturnToCapture(EDT_CPF)
END
HReadSeekFirst(CLIENTE, CPF, EDT_CPF)
IF HFound(CLIENTE) AND CLIENTE.IDCLIENTE <> EDT_IDCliente THEN
    Error("Já existe cliente com este CPF: " + CLIENTE.NOME)
    ReturnToCapture(EDT_CPF)
END

```

Clique em BTN_Salvar

```

ScreenToFile(WIN_Cliente)
IF CLIENTE.DATA_CADASTRO = "" THEN CLIENTE.DATA_CADASTRO = Today()
IF EDT_IDCliente = 0 THEN
    HAdd(CLIENTE)
ELSE
    HModify(CLIENTE)
END
Info("Cliente salvo.")
Close()

```

Projeto ESTOQUE — Documentação técnica: Interfaces

WINDEV 2025 · Update 1 · Gerado em 03/09/2026 · Escopo: janelas, controles, navegação e relatórios

Janela WIN_Menu (janela principal)

Título: «Estoque — Menu principal». Tamanho 900×600. Menu: Cadastros (Clientes, Produtos, Depósitos), Movimento (Nova venda, Consultar vendas), Relatórios (Saldo abaixo do mínimo, Comissão mensal), Sair.

Controle	Tipo	Ação
MENU_Cadastros_Clientes	Opção de menu	Abre WIN_ListaClientes
MENU_Movimento_NovaVenda	Opção de menu	Abre WIN_Venda
MENU_Relatorios_Minimo	Opção de menu	Imprime RPT_SaldoMinimo
MENU_Relatorios_Comissao	Opção de menu	Abre WIN_ParamComissao e imprime RPT_Comissao
STC_Usuario	Static	Mostra gsUsuarioNome na barra de status

Janela WIN_Venda

Título: «Nova venda». Tamanho 860×560. Modal. Aberta por WIN_Menu e por WIN_ListaVendas (botão Nova).

Controle	Tipo	Vínculo	Regras na tela
COMBO_Cliente	Combo box (arquivo CLIENTE, NOME)	VENDA.IDCLIENTE	Obrigatório. Mostra TIPO e LIMITE_CREDITO em STC_Cliente ao selecionar.
COMBO_Deposito	Combo box (DEPOSITO, NOME)	VENDA.IDDEPOSITO	Obrigatório. Padrão: depósito 1.
EDT_Data	Edit (data)	VENDA.DATA	Padrão hoje; não aceita data futura.
COMBO_Produto	Combo box (PRODUTO ativo, DESCRICAO)	—	Busca por código ou descrição.
EDT_Quantidade	Edit (numérico 12.3)	—	Maior que zero.
BTN_AdicionarItem	Botão «Adicionar»	—	Valida estoque antes de incluir a linha.
TABLE_Itens	Tabela memória	—	Colunas: COL_Codigo, COL_Descricao, COL_Quantidade, COL_Precos, COL_TotalItem. Ordenável por descrição. Duplo clique remove a linha.
EDT_PercDesconto	Edit (numérico 5.2)	—	Recalcula ao sair; acima do teto zera e avisa.
EDT_Subtotal, EDT_Desconto, EDT_Total	Edit (moeda), somente leitura	VENDA.SUBTOTAL, DESCONTO, TOTAL	Recalculados por RecalculaTotais.
COMBO_Parcelsas	Combo (1 a 6)	—	Padrão 1.
BTN_Fechar	Botão «Fechar venda»	—	Grava venda, itens, baixa estoque, gera títulos.
BTN_Cancelar	Botão «Cancelar»	—	Fecha sem gravar; pede confirmação se houver itens.

Estados: vazia (sem itens, botão Fechar desabilitado); com itens; erro de estoque (mensagem e foco em EDT_Quantidade); acima do limite de crédito (pergunta Sim/Não).

Janela WIN_Cliente

Título: «Cliente». Tamanho 640×420. Modal. Aberta por WIN_ListaClientes (Novo e Alterar).

Controle	Tipo	Vínculo	Regras na tela
EDT_IDCliente	Edit, somente leitura	CLIENTE.IDCLIENTE	0 em inclusão.
EDT_Nome	Edit (80)	CLIENTE.NOME	Obrigatório.
EDT_CPF	Edit (máscara 999.999.999-99)	CLIENTE.CPF	Validado na saída; único.
EDT_Email	Edit (120)	CLIENTE.EMAIL	Formato de e-mail quando preenchido.
EDT_Cidade, COMBO_UF	Edit, combo (27 UF)	CLIENTE.CIDADE, UF	—
RADIO_Tipo	Radio (Comum, Especial)	CLIENTE.TIPO	Padrão Comum.
EDT_LimiteCredito	Edit (moeda)	CLIENTE.LIMITE_CREDITO	Não negativo.
CBOX_Ativo	Check box	CLIENTE.ATIVO	Padrão marcado.
BTN_Salvar, BTN_Cancelar	Botões	—	—

Janela WIN_ListaClientes

Tabela TABLE_Clientes ligada ao arquivo CLIENTE, colunas NOME, CPF, CIDADE, UF, TIPO, ATIVO. Filtro por nome (EDT_Filtro) e por ativo (CBOX_SoAtivos, padrão marcado). Botões Novo, Alterar, Excluir (pede confirmação; não exclui cliente com venda). Ordenação por qualquer coluna.

Janela WIN_ListaVendas

Tabela sobre a query QRY_VendasPeriodo. Filtros: EDT_DataInicial, EDT_DataFinal (padrão: mês corrente), COMBO_Vendedor. Colunas: IDVENDA, DATA, CLIENTE, VENDEDOR, TOTAL, STATUS. Botão Nova abre WIN_Venda; botão Cancelar venda muda STATUS para C e devolve o estoque (procedure EstornaEstoque). Total do período em STC_TotalPeriodo.

Relatório RPT_SaldoMinimo

Fonte: view VW_SALDO_ABAIXO_MINIMO. Colunas: código, descrição, depósito, quantidade, mínimo. Quebra por depósito com contagem de itens. Retrato, A4.

Relatório RPT_Comissao

Fonte: QRY_ComissaoMensal com parâmetros pMes, pAno. Colunas: vendedor, quantidade de vendas, total vendido, comissão (3% do total das vendas fechadas). Totalizador geral. Paisagem, A4.

Projeto ESTOQUE — Documentação técnica: Queries

WINDEV 2025 · Update 1 · Gerado em 03/09/2026 · Escopo: queries do projeto, parâmetros e onde são usadas

QRY_VendasPeriodo

Parâmetro	Tipo	Obrigatório
pDataInicial	Date	sim
pDataFinal	Date	sim
pVendedor	string	não (vazio = todos)

```
SELECT VENDA.IDVENDA, VENDA.DATA, CLIENTE.NOME AS CLIENTE, VENDA.VENDEDOR, VENDA.TOTAL,
VENDA.STATUS
FROM VENDA
JOIN CLIENTE ON CLIENTE.IDCLIENTE = VENDA.IDCLIENTE
WHERE VENDA.DATA BETWEEN {pDataInicial} AND {pDataFinal}
AND ({pVendedor} = '' OR VENDA.VENDEDOR = {pVendedor})
ORDER BY VENDA.DATA DESC, VENDA.IDVENDA DESC
```

Usada em: WIN_ListaVendas (TABLE_Vendas). Ordenação decrescente por data e id; STATUS C aparece com a linha em cinza.

QRY_SaldoProduto

Parâmetro	Tipo	Obrigatório
pIDProduto	int	sim

```
SELECT DEPOSITO.NOME AS DEPOSITO, SALDO.QUANTIDADE, PRODUTO.ESTOQUE_MINIMO
FROM SALDO
JOIN DEPOSITO ON DEPOSITO.IDDEPOSITO = SALDO.IDDEPOSITO
JOIN PRODUTO ON PRODUTO.IDPRODUTO = SALDO.IDPRODUTO
WHERE SALDO.IDPRODUTO = {pIDProduto}
ORDER BY DEPOSITO.NOME
```

Usada em: WIN_Produto (aba Saldos) e no tooltip de COMBO_Produto em WIN_Venda.

QRY_ComissaoMensal

Parâmetro	Tipo	Obrigatório
pMes	int	sim
pAno	int	sim

```
SELECT VENDA.VENDEDOR, COUNT(VENDA.IDVENDA) AS QTD_VENDAS, SUM(VENDA.TOTAL) AS
TOTAL_VENDIDO,
SUM(VENDA.TOTAL) * 0.03 AS COMISSAO
FROM VENDA
WHERE MONTH(VENDA.DATA) = {pMes} AND YEAR(VENDA.DATA) = {pAno} AND VENDA.STATUS = 'F'
GROUP BY VENDA.VENDEDOR
ORDER BY TOTAL_VENDIDO DESC
```

Usada em: RPT_Comissao. Só vendas fechadas (STATUS = 'F'); canceladas e abertas não contam. Comissão de 3% calculada na query, não na tela.

QRY_TitulosVencidos

Parâmetro	Tipo	Obrigatório
pDataBase	Date	sim

```
SELECT TITULO.IDTITULO, VENDA.IDVENDA, CLIENTE.NOME, TITULO.VENCIMENTO, TITULO.VALOR,
       DATEDIFF({pDataBase}, TITULO.VENCIMENTO) AS DIAS_ATRASO
FROM TITULO
JOIN VENDA ON VENDA.IDVENDA = TITULO.IDVENDA
JOIN CLIENTE ON CLIENTE.IDCLIENTE = VENDA.IDCLIENTE
WHERE TITULO.PAGO_EM IS NULL AND TITULO.VENCIMENTO < {pDataBase}
ORDER BY TITULO.VENCIMENTO
```

Usada em: WIN_Cobranca. Os juros são calculados na tela por CalculaJurosAtraso, com a data de pagamento informada, não pela query.

QRY_ClientesAtivos

```
SELECT IDCLIENTE, NOME, TIPO, LIMITE_CREDITO FROM CLIENTE WHERE ATIVO = 1 ORDER BY NOME
```

Sem parâmetros. Usada em: COMBO_Cliente de WIN_Venda.